

Gasoline Blending

A refinery takes four raw gasolines, blends them, and produces three types of fuel.

Data about raw gasolines

| Raw gas | Octane rating | Available (barrels/day) | Cost (\$/barrel) | Selling price (\$/b) |
|---------|---------------|-------------------------|------------------|----------------------|
| 1 | 68 | 4000 | 31.02 | 36.85 |
| 2 | 86 | 5050 | 33.15 | 36.85 |
| 3 | 91 | 7100 | 36.35 | 38.95 |
| 4 | 99 | 4300 | 38.75 | 38.95 |

Data about blended fuels

| Fuel type | Min octane rating | Selling price (\$/barrel) | Demand pattern |
|-----------|-------------------|---------------------------|--------------------------|
| 1 | 95 | 45.15 | ≤ 10000 barrels/day |
| 2 | 90 | 42.95 | Any amt can be sold |
| 3 | 85 | 40.99 | ≥ 15000 barrels/day |

```

MAX= @SUM(RAW(I): (PRAW(I)-COST(I))*Y(I)
        + @SUM(FUEL(J): (PRICE(J)-COST(I))*X(I,J))));

! minimum octane requirement for each fuel;
@FOR(FUEL(J):
    @SUM(RAW(I): OCTANE(I)*X(I,J)) >=
        MINOCT(J)*@SUM(RAW(I): X(I,J));
);

! maximum production of fuel #1;
@SUM(RAW(I): X(I,1)) <= DEMAND(1);
!minimum production of fuel #3;
@SUM(RAW(I): X(I,3)) >= DEMAND(3);

! availability of raw gasolines;
@FOR(RAW(I):
    @SUM(FUEL(J): X(I,J)) + Y(I) <= AVAIL(I);
);
END

```

The LINGO model:

```

MODEL: ! GASOLINE BLENDING PROBLEM;

SETS:
  RAW/1..4/:OCTANE,AVAIL,COST,PRAW,Y;
  FUEL/1..3/:MINOCT,PRICE,DEMAND;
  BLEND(RAW,FUEL):X;
ENDSETS

DATA:
!  attributes of raw gasolines;
OCTANE= 68 86 91 95;
AVAIL= 4000 5050 7100 4300;
COST= 31.02 33.15 36.35 38.75;
PRAW= 36.35 36.35 38.95 38.95; ! selling price of raw gas;
!  attributes of blended fuel types;
MINOCT= 95 90 85;
PRICE= 45.15 42.95 40.99;
DEMAND= 10000 0 15000;
ENDDATA

```

The LINDO model generated by LINGO:

```

MAX      14.13 X( 1, 1) + 11.93 X( 1, 2) + 9.97 X( 1, 3) + 12 X( 2, 1)
        + 9.8 X( 2, 2) + 7.84 X( 2, 3) + 8.8 X( 3, 1) + 6.6 X( 3, 2)
        + 4.64 X( 3, 3) + 6.4 X( 4, 1) + 4.2 X( 4, 2) + 2.24 X( 4, 3)
        + 5.33 Y( 1) + 3.2 Y( 2) + 2.6 Y( 3) + .2 Y( 4)
SUBJECT TO
2]- 27 X( 1, 1) - 9 X( 2, 1) - 4 X( 3, 1) >= 0
3]- 22 X( 1, 2) - 4 X( 2, 2) + X( 3, 2) + 5 X( 4, 2) >= 0
4]- 17 X( 1, 3) + X( 2, 3) + 6 X( 3, 3) + 10 X( 4, 3) >= 0
5] X( 1, 1) + X( 2, 1) + X( 3, 1) + X( 4, 1) <= 10000
6] X( 1, 3) + X( 2, 3) + X( 3, 3) + X( 4, 3) >= 15000
7] X( 1, 1) + X( 1, 2) + X( 1, 3) + Y( 1) <= 4000
8] X( 2, 1) + X( 2, 2) + X( 2, 3) + Y( 2) <= 5050
9] X( 3, 1) + X( 3, 2) + X( 3, 3) + Y( 3) <= 7100
10] X( 4, 1) + X( 4, 2) + X( 4, 3) + Y( 4) <= 4300
END

```

The solution found by LINGO:

Primal solution

| Variable | Value | Reduced Cost |
|----------|----------|--------------|
| X(1, 3) | 2820.370 | 0.0000000 |
| X(2, 3) | 5050.000 | 0.0000000 |
| X(3, 3) | 7100.000 | 0.0000000 |
| X(4, 1) | 4270.370 | 0.0000000 |
| X(4, 3) | 29.62963 | 0.0000000 |
| Y(1) | 1179.630 | 0.0000000 |

Dual solution

| Row | Slack or Surplus | Dual Price |
|-----|------------------|------------|
| 2 | 0.0000000 | -0.3259259 |
| 3 | 0.0000000 | -0.3000000 |
| 4 | 0.0000000 | -0.3259259 |
| 5 | 5729.630 | 0.0000000 |
| 6 | 0.0000000 | -0.9007407 |
| 7 | 0.0000000 | 5.3300000 |
| 8 | 0.0000000 | 9.066667 |
| 9 | 0.0000000 | 7.496296 |
| 10 | 0.0000000 | 6.400000 |

| Row | Righthand Side Ranges | | |
|-----|-----------------------|--------------------|--------------------|
| | Current RHS | Allowable Increase | Allowable Decrease |
| 2 | 0.0 | 0.0 | 31850.00 |
| 3 | 0.0 | 0.0 | 25951.85 |
| 4 | 0.0 | 76150.00 | 800.0000 |
| 5 | 10000.00 | INFINITY | 5729.630 |
| 6 | 15000.00 | 3185.000 | 47.05882 |
| 7 | 4000.000 | INFINITY | 1179.630 |
| 8 | 5050.000 | 44.44444 | 3538.889 |
| 9 | 7100.000 | 34.78261 | 5013.043 |
| 10 | 4300.000 | 5729.630 | 4270.370 |

Range analysis

Ranges in which the basis is unchanged:

| Variable | Objective Coefficient Ranges | | |
|----------|------------------------------|--------------------|--------------------|
| | Current Coefficient | Allowable Increase | Allowable Decrease |
| Y(1) | 5.330000 | 6.600000 | 2.432000 |
| Y(2) | 3.200000 | 5.866667 | INFINITY |
| Y(3) | 2.600000 | 4.896296 | INFINITY |
| Y(4) | 0.2000000 | 6.200000 | INFINITY |
| X(1, 1) | 14.13000 | INFINITY | 0.0 |
| X(1, 2) | 11.93000 | 3.080000 | 2.566667 |
| X(1, 3) | 9.970000 | 0.0 | 8.800000 |
| X(2, 1) | 12.00000 | 0.0 | INFINITY |
| X(2, 2) | 9.800000 | 0.4666667 | INFINITY |
| X(2, 3) | 7.840000 | INFINITY | 0.0 |
| X(3, 1) | 8.800000 | 0.0 | INFINITY |
| X(3, 2) | 6.600000 | 0.5962963 | INFINITY |
| X(3, 3) | 4.640000 | INFINITY | 0.0 |
| X(4, 1) | 6.400000 | INFINITY | 0.0 |
| X(4, 2) | 4.200000 | 0.7000000 | INFINITY |
| X(4, 3) | 2.240000 | 0.0 | INFINITY |